

PROC PRINT - the Granddaddy of all Procedures, Enhanced and Still Going Strong! Paper DS09

David Franklin
TheProgrammersCabin.com, Independent SAS Consultant

On this day in 1959, Bonanza, the first
regularly-scheduled TV program presented in
color, premiered

Paper DS09, NESUG2011 : David Franklin

Introduction

- The PRINT procedure, or PROC PRINT, has been around since SAS first began and is considered one of the granddaddy procedures (was in the original releases of SAS)
- Although replaced in part by the REPORT procedure, there is still a lot you can do with it.
- This paper looks at first a simple dump of data, then dresses it up with the use of statements like the BY and ID statements to publication ready output, cranked up a notch to demonstrate how PROC PRINT ODS enhancements can be used to produce HTML (with graphics and links), RTF and PDF (with bookmarks).
- CAN DO MORE THAN JUST A DUMP OF YOUR DATA

A Simple Dump

```
proc print data=SheepNumbers;  
  var Country Year Number;  
  title2 "(Basic Output)";  
run;
```

Sheep Numbers in Select Countries, By Year
(Basic Output)

Obs	Country	Year	Number
1	Australia	2007	85711187
2	Australia	2006	91028408
3	Australia	2005	91028408
4	Australia	2004	91028408
5	Australia	2003	91028408
6	Australia	2002	91028408
7	Australia	2001	91028408
8	Australia	2000	91028408
9	Australia	1999	91028408
10	Australia	1998	91028408
11	Australia	1997	91028408
12	Australia	1996	91028408
13	Australia	1995	91028408
14	Australia	1994	91028408
15	Australia	1993	91028408
16	Australia	1992	91028408
17	Australia	1991	91028408
18	Australia	1990	91028408
19	Australia	1989	91028408
20	Australia	1988	91028408
21	United Kingdom	2005	35253048
22	United States of America	2007	6165000
23	United States of America	2006	6230000
24	United States of America	2005	6135000

Note: Numbers for India are FAO Estimate

Source: UNData, 03Nov2010


Paper DS09, NESUG2011 : David Franklin

The NOOBS and LABEL options

```
proc print data=SheepNumbers noobs label;  
  var Country Year Number;  
  label Country='Country'  
        Year='Year'  
        Number='Reported Number';  
  title2 "(Something a Little Better)";  
run;
```

Sheep Numbers in Select Countries, By Year
(Something a Little Better)

Country	Year	Reported Number
Australia	2007	85711187
Australia	2006	91028408
United States of America	2006	6230000
United States of America	2005	6135000



Note: Numbers for India are FAO Estimate
Source: UNData, 03Nov2010

BY AND ID STATEMENTS

- Adds a type of 'Grouping'

```
proc print data=SheepNumbers label;  
  by Country; id Country; var Year Number;  
  format Number comma12.;  
  label Country='Country' Year='Year' Number='Reported Number';  
  title2 "(Something even better, with BY and ID)";  
run;
```

Sheep Numbers in Select Countries, By Year
(Something even better, with BY and ID)

Country	Year	Reported Number
New Zealand	2007	38,460,477
	2006	40,098,191
	2005	39,879,660

A Line Underneath the Heading I

- Unlike the REPORT procedure, a line underneath the column headings does not appear in this mode, but does appear when using ODS destinations as we shall see later. Given all the advancements with PROC PRINT it would be nice to have that option outside ODS.
- In order for a line to appear underneath the column headings, some post-processing needs to be done in the following order:
 - Make a line underneath the column headings – does not need to be all the way across except for the first and last variables (see below of an example). I find it useful to use a character not used elsewhere to set the underline character at this stage.
 - When the file is read in after the PROC PRINT call, if the line carries the characters for used for the line, find the first and last positions.
 - Fill the text between the first and last characters of the line with the character that is going to be the line character.

A Line Underneath the Heading III

```
data _null_ ;
  length _txt $255;
  infile 'E:\NESUG\2011\ODS\POSTPRC1.txt' sharebuffers
  length=len;
  file 'E:\NESUG\2011\ODS\POSTPRC1.txt';
  input _txt $varying255. len;
  if index(_txt,'£') then do;
    startpos= index(_txt,'£');
    do k= startpos to len;
      if substr(_txt,_k)='£' then _endpos=_k;
    end;
    substr(_txt,_startpos,_endpos)=repeat('-',_endpos-
_startpos);
  end;
  put _txt $varying255. len;
run;
```

- Because the INFILE statement uses the SHAREBUFFERS option, the INFILE and FILE statements must use the same file name. Also used is the LEN option in the INFILE statement to tell how long the incoming string is and use this in the INPUT statement.
- Only if a '£' is found in the incoming line, will the next step proceed, where we will first find the first and last position of the '£' characters in the string, and then populate that string filling in the gaps with the replacement character, which in this case is the '-' character.

A Line Underneath the Heading IV

Sheep Numbers in Select Countries, By Year
(Something even better, with what looks like the HEADLINE option in PROC REPORT)

Country	Year	Reported Number
Australia	2007	85,711,187
	2006	91,028,408
	2005	101,124,891
Canada	2007	879,100
	2006	893,800
	2005	977,600
China	2007	146,018,203
	2006	151,337,202
	2005	152,035,223
India	2007	64,269,000
	2006	63,558,000
	2005	62,854,000
United Kingdom	2007	33,946,000

SUM and N Statements

```
proc print data=SheepNumbers label n='Country Count=' 'Total Count=';
  title2 "(Output using the SUM Statement)";
  by Country; id Country; var Year Number;
  format Number comma18.;
  label Country='Country' Year='Year' Number='Reported Number';
  sum number; ** What variables to SUM;
run;
```

- N statement, as used here, will put out the number of observations found in each BY variable and give the label 'Country Count' and 'Total Count' for the total number of observations.

Sheep Numbers in Select Countries, By Year
(Output using the SUM Statement)

Country	Year	Reported Number
Australia	2007	85,711,187
	2006	91,028,408
	2005	101,124,891
-----	-----	-----
Australia		277,864,486

Country Count = 3

Paper DS09, NESUG2011 : David Franklin

ODS HTML

```
ods listing close;
ods html file='E:\NESUG\2011\ods\htmlexample.html' style=custom;
proc print data=SheepNumbers label split='!';
  by Country; id Country;
  var Year /style=[cellwidth=0.5in just=c];
  var Number /style=[cellwidth=1.0in];;
  format Number comma12.;
  label Country='Country' Year='Year' Number='Reported!Number';
  title2 "(HTML Output, Using Custom Template)";
run;
ods html close; ods listing;
run;
```

Country	Year	Reported Number
Australia	2005	101,124,891
	2006	91,028,408
	2007	85,711,187
Canada	2005	977,600
	2006	893,800
	2007	879,100
China	2005	152,035,223

ODS Options in PROC PRINT, 101

- It must be noted here that the use of the STYLE= options have been around since ODS started:
 - BYLABEL -- the label for the BY variable on the line containing the SUM totals
 - DATA -- the cells of all columns
 - GRANDTOTAL -- the SUM line containing the grand totals for the whole report
 - HEADER -- all column headings
 - N -- N= table and contents
 - OBS -- the data in the OBS column
 - OBSHEADER -- the header of the OBS column
 - TABLE -- the structural part of the report - that is, the underlying table used to set things like the width of the border and the space between cells
 - TOTAL -- the SUM line containing totals for each BY group
- Style attributes that are available include:
 - BORDERWIDTH= CELLHEIGHT= CELLPADDING= CELLSPACING=
 - CELLWIDTH= FONT= FONT_FACE= FONT_SIZE= FONT_STYLE=
 - FONT_WEIGHT= JUST= POSTIMAGE= PRETEXT= RULES= VJUST=
- There are others – refer to the PROC PRINT documentation.

ODS RTF

```
ods rtf
```

```
file='E:\NESUG\2011\ods\rtfexample.rtf'
```

```
style=custom;
```

Sheep Numbers in Select Countries, By Year
(RTF Output, Using Custom Template)

Country	Year	Reported Number
Australia	2005	101,124,891
	2006	91,028,408
	2007	85,711,187
Canada	2005	977,600
	2006	893,800
	2007	879,100
China	2005	152,035,223
	2006	151,337,202
	2007	146,018,203
India	2005	62,854,000
	2006	63,558,000
	2007	64,269,000
New Zealand	2005	39,879,660
	2006	40,098,191
	2007	38,460,477
South Africa	2005	25,334,000
	2006	24,982,996
	2007	25,082,100
United Kingdom	2005	35,253,048
	2006	34,722,000
	2007	33,946,000
United States of America	2005	6,135,000
	2006	6,230,000
	2007	6,165,000

Note: Numbers for India are FAO Estimate
Source: UNData, 03Nov2010

ODS PDF

```
ods pdf  
file='E:\NESUG\2011\ods\pdfexample.pdf'  
style=custom;
```

- IN ODS, no option to add a blank line between groups – that can be done manually by creating a blank record at the end of each country with a year value >2007 and formatting the value to blank.
- The following code will create a blank line!

```
proc format;  
  value yearf 2009=' '  
data SheepNumbers;  
  set SheepNumbers;  
  by Country Year;  
  output SheepNumbers;  
  if last.Country then do;  
    call missing(Number);  
    Year=2009;  
    output;  
  end;  
  format Year yearf.;  
run;
```

Sheep Numbers in Select Countries, By Year
(PDF Output, Using Custom Template)

Country	Year	Reported Number
Australia	2005	101,124,891
	2006	91,028,408
	2007	85,711,187
Canada	2005	977,600
	2006	893,800
	2007	879,100
China	2005	152,035,223
	2006	151,337,202
	2007	146,018,203
India	2005	62,854,000
	2006	63,558,000
	2007	64,269,000
New Zealand	2005	39,879,660
	2006	40,098,191
	2007	38,460,477
South Africa	2005	25,334,000
	2006	24,982,996
	2007	25,082,100
United Kingdom	2005	35,253,048
	2006	34,722,000
	2007	33,946,000
United States of America	2005	6,135,000
	2006	6,230,000
	2007	6,165,000


ADDING LINKS AND GRAPHICS IN HTML I

```
ods listing close;
ods html file='E:\NESUG\2011\ods\htmlexamplejpg.html' style=custom;
proc print data=SheepNumbers label split='!'
      STYLE=[PREIMAGE='E:\NESUG\2011\ods\border.jpg'
            PREHTML='<a href="http://en.wikipedia.org/wiki/Sheep">Link
to more about Sheep</a>'];
  by Country;
  id Country;  ** Acts like a "GROUPING" variable;
  var Year /style=[cellwidth=0.5in just=c];
  var Number /style=[cellwidth=1.0in];;
  format Number comma12.;
  label Country='Country'
         Year='Year'
         Number='Reported!Number';
  title2 "(HTML Output with Graphic and Link, Using Custom Template)";
run;
ods html close;
ods listing;
run;
```

ADDING LINKS AND GRAPHICS IN HTML II

Sheep Numbers in Select Countries, By Year
(HTML Output with Graphic and Link, Using Custom Template)

[Link to more about Sheep](#)




Country	Year	Reported Number
Australia	2005	101,124,891
	2006	91,028,408
	2007	85,711,187
Canada	2005	977,600
	2006	893,800
	2007	879,100
China	2005	152,035,223
	2006	151,337,202
	2007	146,018,203
India	2005	62,854,000
	2006	63,558,000
	2007	64,269,000
New Zealand	2005	39,879,660
	2006	40,098,191
	2007	38,460,477

ADDING A GRAPHIC IN RTF AND PDF

```
ods listing close;
ods rtf file='E:\NESUG\2011\ods\rtfexamplejpg.rtf' style=custom;
proc print data=SheepNumbers label split='!'
      STYLE=[PREIMAGE='E:\NESUG\2011\ods\border.jpg'];
  by Country; id Country;
  var Year /style=[cellwidth=0.5in just=c];
  var Number /style=[cellwidth=1.0in];;
  format Number comma12.;
  label Country='Country' Year='Year' Number='Reported!Number';
  title2 "(RTF Output with Graphic, Using Custom Template)";
run;
ods rtf close;
ods listing;
run;
```

- Similar in ODS PDF, just change ODS RTF to ODS PDF

Sheep Numbers in Select Countries, By Year
(RTF Output with Graphic, Using Custom Template)




<u>Country</u>	<u>Year</u>	<u>Reported Number</u>
Australia	2005	101,124,891
	2006	91,028,408
	2007	85,711,187

BOOKMARKS AND PDF I

```
ods listing close;
ods pdf file='E:\NESUG\2011\ods\pdfexamplelink.pdf'
  style=custom;
ods escapechar="!";
options byline;
proc sort data=SheepNumbers;
  by Country Year;
run;
ods proclabel="Listing of Sheep Data";
proc print data=SheepNumbers label split='!' contents='';
  by Country; id Country;
  var Year /style=[cellwidth=0.5in just=c];
  var Number /style=[cellwidth=1.0in];;
  format Number comma12.;
  label Country='Country' Year='Year' Number='Reported!
Number';
  title2 "(PDF Output with Bookmarks, Using Custom Template)";
run;
ods pdf close;
ods listing;
run;
```

BOOKMARKS AND PDF II

- ODS PROCLABEL replaces the default text associated with the bookmark at the first level
- The CONTENTS option in the PROC PRINT statement sets text at the second level in the bookmark to missing



Sheep Numbers in Select Countries, By Year
(PDF Output with Bookmarks, Using Custom Template)

<u>Country</u>	<u>Year</u>	<u>Reported Number</u>
Australia	2005	101,124,891
	2006	91,028,408
	2007	85,711,187

Conclusion

- The PRINT procedure is not just something used to “dump” data from a dataset to a text listing file.
- During the tour we have seen how simple things like NOOBS and LABEL options, and BY and ID statements make the output more presentable.
- An example of how we can post process a file was also shown.
- Next on our journey was a brief introduction into how ODS HTML/RTF/PDF has enhanced the PRINT procedure.
- The PRINT procedure may be one of the oldest procedures in the SAS stable but it still has life today as a serious option when considering what procedure to use to produce your output.

Questions and Contact Information



Questions?

Contact Information

David Franklin

TheProgrammersCabin.com

16 Roberts Road, Litchfield, NH 03052

Phone: 603-275-6809

Email: dfranklin@TheProgrammersCabin.com

Web: <http://www.TheProgrammersCabin.com>

LinkedIn: <http://www.linkedin.com/in/davidfranklinnh>



Paper DS09, NESUG2011 : David Franklin

Acknowledgement

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are registered trademarks or trademarks of their respective companies.